

FILE S1

Implementation of the univariate mixed linear model in WinBUGS

```

model
{
# Loop over all individuals for inference of error (co)variance, assumed to be i.i.d.
for( i in 1 : nind ) {
  yr [i] ~ dnorm(mu[i], omega.err)
}

# Loop over all pedigree members for inference of genetic values and (co)variance
for ( i in 1 : nind ){
  mu[i] <- beta[X[i]] + gen[i]
  for(j in nst[i]:nel[i]){ # looping over all non-zero elements in weight matrix
    r[j] <- na[j]*gen[nid[j]]
  }
  par.gen[i] <- sum(r[nst[i]:nel[i]]) # calculating mean for pedigree member i
  gen[i] ~ dnorm(par.gen[i] , prec.gen[i] )
}

# Specification of prior distributions

# systematic effect assuming a flat prior, nlevel levels of systematic effect
for (i in 1:nlevel){
  beta[i] ~ dnorm(0,1.0E-6)
}

# variance components assuming a uniform distribution as prior (see Waldmann 2009)
omega.err <- 1 / VE
sigma.err ~ dunif(0, sd.u.err)
VE <- sigma.err * sigma.err
omega.gen <- 1 / VG
sigma.gen ~ dunif(0, sd.u.gen)
VG <- sigma.gen * sigma.gen

# precision for genetic effect

```

```
for (i in 1:nind){  
  prec.gen[ i ] <-omega.gen/wvar[i]  
}
```

```
# Specification of functions of model parameters of inferential interest
```

```
h2 <- VG/(VG + VE) # Heritability
```

```
}
```

REFERENCES

Waldmann, P. 2009 Easy and flexible Bayesian inference of quantitative genetic parameters. *Evolution* **63**: 1640-1643.